

Stata 10 Tutorial 5

TOPIC: Hypothesis Testing of Individual Regression Coefficients: Two-Tail t-tests, Two-Tail F-tests, and One-Tail t-tests

DATA: `auto1.dta` (a *Stata*-format data file created in *Stata Tutorial 1*)

TASKS: *Stata Tutorial 5* has three primary purposes: (1) to demonstrate how to compute *two-tail t-tests* of individual regression coefficients and the corresponding **p-values** of the calculated t-statistics; (2) to introduce you to *two-tail F-tests* in linear regression models and to the **test** command, a post-estimation command used to compute F-tests of linear equality restrictions on regression coefficients; and (3) to give you some practice in computing *one-tail t-tests* and the corresponding **p-values** of the calculated t-statistics.

- The *Stata commands* that constitute the primary subject of this tutorial are:

regress	(So what's new?)
scalar	Defines the contents of scalar variables.
scalar list	Lists names and values of currently-defined scalar variables.
display	Computes and displays values of algebraic expressions.
test	Used after OLS estimation to compute <i>two-tail F-tests</i> of coefficient equality restrictions.
return list	Lists all temporarily-saved results of the test command.
lincom	Used after OLS estimation to compute <i>two-tail t-tests</i> of individual regression coefficients.

- The *Stata statistical functions* introduced in this tutorial are:

ttail (<i>df</i> , <i>t</i> ₀)	Computes <i>p-values</i> of <i>calculated t-statistics</i> .
invttail (<i>df</i> , <i>p</i>)	Computes <i>right-tail critical values</i> of <i>t-distributions</i> .
Ftail (<i>df</i> ₁ , <i>df</i> ₂ , <i>F</i> ₀)	Computes <i>p-values</i> of <i>calculated F-statistics</i> .
invFtail (<i>df</i> ₁ , <i>df</i> ₂ , α)	Computes <i>right-tail critical values</i> of <i>F-distributions</i> .

NOTE: *Stata* commands are *case sensitive*. All *Stata command names* must be typed in the Command window in **lower case letters**.

□ Preparing for Your *Stata* Session

Before beginning your *Stata* session, use Windows Explorer to copy the *Stata*-format dataset **auto1.dta** to the *Stata working directory* on the C:-drive or D:-drive of the computer at which you are working.

- **On the computers in Dunning 350**, the default *Stata* working directory is usually **C:\data**.
- **On the computers in MC B111**, the default *Stata* working directory is usually **D:\courses**.

□ Start Your *Stata* Session

To start your *Stata* session, double-click on the ***Stata 10* icon** in the Windows desktop.

After you double-click the ***Stata 10* icon**, you will see the now familiar screen of four *Stata* windows.

□ Record Your *Stata* Session -- log using

To record your *Stata* session, including all the *Stata* commands you enter and the results (output) produced by these commands, make a **.log** file named **351tutorial5.log**. To open (begin) the **.log** file **351tutorial5.log**, enter in the Command window:

```
log using 351tutorial5.log
```

This command opens a file called **351tutorial5.log** in the current *Stata* working directory. Remember that once you have opened the **351tutorial5.log** file, a copy of all the commands you enter during your *Stata* session and of all the results they produce is recorded in that **351tutorial5.log** file.

An alternative way to open the **.log** file **351tutorial5.log** is to click on the **Log** button; click on **Save as type:** and select **Log (*.log)**; click on the **File name:** box and type the file name **351tutorial5**; and click on the **Save** button.

□ Loading a *Stata*-Format Dataset into *Stata* -- use

Load, or read, into memory the dataset you are using. To load the *Stata*-format data file **auto1.dta** into memory, enter in the Command window:

```
use auto1
```

This command loads into memory the *Stata*-format dataset **auto1.dta**.

□ Familiarize Yourself with the Current Dataset

To familiarize (or re-familiarize) yourself with the contents of the current dataset, type in the Command window the following commands:

```
describe
summarize
codebook
```

□ Computing a Simple OLS Regression Equation (again!) -- regress

- Most of the *Stata* commands covered in this tutorial are post-estimation commands -- that is, they are used after a model estimation command such as a **regress** command. We therefore need an OLS sample regression equation to illustrate the use of these post-estimation commands.
- So estimate again by OLS the simple linear regression model given by the PRE

$$\text{price}_i = \beta_0 + \beta_1 \text{weight}_i + u_i. \quad (1)$$

Enter in the Command window the following commands:

```
regress price weight
ereturn list
```

- Recall from *Stata Tutorial 3* that selected results of the **regress** command are temporarily saved in **e()** functions. For later use in this tutorial, it will be convenient to save the three terms in the OLS decomposition equation (and

their degrees of freedom) and the number of sample observations N . To do this, enter the following **scalar** commands:

```
scalar N = e(N)
scalar ESS = e(mss)
scalar dfESS = e(df_m)
scalar RSS = e(rss)
scalar dfrSS = e(df_r)
scalar TSS = ESS + RSS
scalar dftSS = N - 1
scalar list N ESS dfESS RSS dfrSS TSS dftSS
```

- You will also want to save for later use the sample value of the F-statistic generated by the **regress** command. Enter the commands:

```
scalar Fstat = e(F)
scalar list Fstat dfESS dfrSS
```

□ Computing two-tail t-tests of equality restrictions on individual regression coefficients -- scalar

A very common type of hypothesis test in applied econometrics consists of testing whether a regression coefficient is equal to some specified value. *Two-tail tests* of such hypotheses take the following general form:

the **null hypothesis** is $H_0: \beta_1 = b_1$

the **alternative hypothesis** is $H_1: \beta_1 \neq b_1$

where b_1 is a specified constant.

The *t-statistic* for $\hat{\beta}_1$ takes the general form:

$$t(\hat{\beta}_1) = \frac{\hat{\beta}_1 - \beta_1}{\widehat{se}(\hat{\beta}_1)} \sim t[N - 2]. \quad (2)$$

The *calculated t-statistic* for testing $H_0: \beta_1 = b_1$ against $H_1: \beta_1 \neq b_1$ is obtained by setting $\beta_1 = b_1$ in formula (2) for $t(\hat{\beta}_1)$:

$$t_0(\hat{\beta}_1) = \frac{\hat{\beta}_1 - b_1}{\widehat{se}(\hat{\beta}_1)} \sim t[N - 2] \text{ under the null hypothesis } H_0: \beta_1 = b_1. \quad (3)$$

A series of **scalar** commands can be used to see how the required t-statistic (3) is calculated. We illustrate with four hypothesis tests on OLS regression equation (1).

Test 1: a two-tail t-test of $H_0: \beta_1 = 0$ vs. $H_1: \beta_1 \neq 0$

- The null and alternative hypotheses are:

$$H_0: \beta_1 = 0 \text{ versus } H_1: \beta_1 \neq 0$$

- The following **scalar** commands name, save and list the values of the *OLS coefficient estimate* $\hat{\beta}_1$, its *estimated standard error* $\widehat{se}(\hat{\beta}_1)$, and its *estimated variance* $\widehat{var}(\hat{\beta}_1)$. Enter the commands:

```
scalar b1 = _b[weight]
scalar seb1 = _se[weight]
scalar varb1 = seb1^2
scalar list b1 seb1 varb1
```

- The *sample value of the t-statistic* for testing $H_0: \beta_1 = 0$ against $H_1: \beta_1 \neq 0$ is obtained either by setting $\beta_1 = 0$ in formula (2) for $t(\hat{\beta}_1)$, or by setting $b_1 = 0$ in formula (3) for $t_0(\hat{\beta}_1)$:

$$t_0(\hat{\beta}_1) = \frac{\hat{\beta}_1 - 0}{\widehat{se}(\hat{\beta}_1)} = \frac{\hat{\beta}_1}{\widehat{se}(\hat{\beta}_1)} \sim t[N - 2] \text{ under } H_0: \beta_1 = 0. \quad (4)$$

To calculate and list the value of $t_0(\hat{\beta}_1)$, enter the commands:

```
scalar tbleq0 = b1/seb1
scalar list b1 seb1 tbleq0
```

Test 2: a two-tail t-test of $H_0: \beta_0 = 0$ vs. $H_1: \beta_0 \neq 0$

- The null and alternative hypotheses are:

$$H_0: \beta_0 = 0 \text{ versus } H_1: \beta_0 \neq 0$$

- The *t*-statistic for $\hat{\beta}_0$ takes the general form:

$$t(\hat{\beta}_0) = \frac{\hat{\beta}_0 - \beta_0}{\hat{se}(\hat{\beta}_0)} \sim t[N - 2]. \quad (5)$$

The *calculated t-statistic* for testing $H_0: \beta_0 = 0$ against $H_1: \beta_0 \neq 0$ is obtained by setting $\beta_0 = 0$ in formula (5) for $t(\hat{\beta}_0)$:

$$t_0(\hat{\beta}_0) = \frac{\hat{\beta}_0 - 0}{\hat{se}(\hat{\beta}_0)} = \frac{\hat{\beta}_0}{\hat{se}(\hat{\beta}_0)} \sim t[N - 2] \text{ under } H_0: \beta_0 = 0. \quad (6)$$

- The following **scalar** commands name, save and list the values of the *OLS coefficient estimate* $\hat{\beta}_0$, its *estimated standard error* $\hat{se}(\hat{\beta}_0)$, and its *estimated variance* $\hat{var}(\hat{\beta}_0)$. Enter the commands:

```
scalar b0 = _b[_cons]
scalar seb0 = _se[_cons]
scalar varb0 = seb0^2
scalar list b0 seb0 varb0
```

- The *sample value of the t-statistic* for testing $H_0: \beta_0 = 0$ against $H_1: \beta_0 \neq 0$ is given by formula (6) above. To calculate and list the value of $t_0(\hat{\beta}_0)$, enter the commands:

```
scalar tb0eq0 = b0/seb0
scalar list b0 seb0 tb0eq0
```

Test 3: a two-tail t-test of $H_0: \beta_1 = 1$ vs. $H_1: \beta_1 \neq 1$

- The null and alternative hypotheses are:

$$H_0: \beta_1 = 1 \text{ versus } H_1: \beta_1 \neq 1$$

- The *sample value of the t-statistic* for testing $H_0: \beta_1 = 1$ against $H_1: \beta_1 \neq 1$ is obtained either by setting $\beta_1 = 1$ in formula (2) for $t(\hat{\beta}_1)$, or by setting $b_1 = 1$ in formula (3) for $t_0(\hat{\beta}_1)$:

$$t_0(\hat{\beta}_1) = \frac{\hat{\beta}_1 - 1}{\widehat{se}(\hat{\beta}_1)} \sim t[N - 2] \text{ under } H_0: \beta_1 = 1. \quad (7)$$

To calculate and list the value of $t_0(\hat{\beta}_1)$ in (7), enter the commands:

```
scalar tbleq1 = (b1 - 1)/seb1
scalar list b1 seb1 tbleq1
```

- An alternative – and simpler – way to compute a two-tail t-test of the null hypothesis $H_0: \beta_1 = 1$ against the alternative hypothesis $H_1: \beta_1 \neq 1$ is to use the *Stata lincom* command. As its name suggests, the **lincom** command computes the value of any linear combination of coefficient estimates, its estimated standard error, its t-statistic and two-tail p-value, and its two-sided 95 percent confidence interval. In the present case, the relevant linear combination is $\hat{\beta}_1 - 1$, the numerator of the required t-statistic for testing $H_0: \beta_1 = 1$ against $H_1: \beta_1 \neq 1$. Note that testing $H_0: \beta_1 = 1$ against $H_1: \beta_1 \neq 1$ is obviously equivalent to testing $H_0: \beta_1 - 1 = 0$ against $H_1: \beta_1 - 1 \neq 0$. Enter the command:

```
lincom _b[weight] - 1
```

Note that **_b[weight]** denotes the OLS slope coefficient estimate $\hat{\beta}_1$ for the regressor **weight** in the most recently executed estimation command. Carefully inspect the results of this **lincom** command, and compare them with the sample value of the t-statistic $t_0(\hat{\beta}_1 - 1)$ contained in the scalar **tbleq1**. State the inference you would draw from this two-tail t-test.

Test 4: a two-tail t-test of $H_0: \beta_1 = 2$ vs. $H_1: \beta_1 \neq 2$

- The null and alternative hypotheses are:

$$H_0: \beta_1 = 2 \text{ versus } H_1: \beta_1 \neq 2$$

- The *sample value of the t-statistic* for testing $H_0: \beta_1 = 2$ against $H_1: \beta_1 \neq 2$ is obtained either by setting $\beta_1 = 2$ in formula (2) for $t(\hat{\beta}_1)$, or by setting $b_1 = 2$ in formula (3) for $t_0(\hat{\beta}_1)$:

$$t_0(\hat{\beta}_1) = \frac{\hat{\beta}_1 - 2}{\widehat{se}(\hat{\beta}_1)} \sim t[N - 2] \text{ under } H_0: \beta_1 = 2. \quad (8)$$

To calculate and list the value of $t_0(\hat{\beta}_1)$ in (8), enter the commands:

```
scalar tbleq2 = (b1 - 2)/seb1
scalar list b1 seb1 tbleq2
```

- As we first saw in *Test 3* above, an alternative – and simpler – way to compute a two-tail t-test of the null hypothesis $H_0: \beta_1 = 2$ against the alternative hypothesis $H_1: \beta_1 \neq 2$ is to use the *Stata* **lincom** command to perform an equivalent test of $H_0: \beta_1 - 2 = 0$ against $H_1: \beta_1 - 2 \neq 0$. In this case, the *Stata* **lincom** command computes the value of the linear combination $\hat{\beta}_1 - 2$, the numerator of the required t-statistic for testing $H_0: \beta_1 = 2$ against $H_1: \beta_1 \neq 2$. Enter the command:

```
lincom _b[weight] - 2
```

Carefully inspect the results of this **lincom** command, and compare them with the sample value of the t-statistic $t_0(\hat{\beta}_1 - 2)$ contained in the scalar **tbleq2**. State the inference you would draw from this two-tail t-test.

□ Computing critical values of t-distributions and p-values for t-statistics

Basic Syntax: The *Stata 10* statistical functions for the t-distribution are **ttail**(*df*, *t*₀) and **invttail**(*df*, *p*).

- **ttail**(*df*, *t*₀) computes the **right-tail (upper-tail) p-value of a t-statistic** that has degrees of freedom *df* and calculated sample value *t*₀. It returns the probability that $t > t_0$, i.e., the value of $\Pr(t > t_0)$.
- **invttail**(*df*, *p*) computes the **right-tail critical value of a t-distribution** with degrees of freedom *df* and **probability level** *p*. Let α denote the chosen significance level of the test. For two-tail t-tests, set $p = \alpha/2$. For one-tail t-tests, set $p = \alpha$.
- If **ttail**(*df*, *t*₀) = *p*, then **invttail**(*df*, *p*) = *t*₀.

Usage: The statistical functions **ttail**(*df*, *t*₀) and **invttail**(*df*, *p*) must be used with *Stata 10* commands such as **display**, **generate**, **replace**, or **scalar**; they cannot be used by themselves.

- For example, simply typing **ttail(60, 2.0)** will produce an error message. Instead, to obtain the **right-tail p-value** for a calculated t-statistic that equals 2.0 and has the t-distribution with 60 degrees of freedom, enter the **display** command:

```
display ttail(60, 2.0)
```

- Although the statistical function **ttail**(*df*, *t*₀) computes the value of $\Pr(t > t_0)$, the right-tail p-value of the calculated t-statistic *t*₀, it can easily be used to compute the **two-tail p-value of *t*₀**.

The **two-tail p-value of *t*₀** is defined as:

$$\text{two-tail p-value of } t_0 = \Pr(|t| \geq |t_0|) = \Pr(t \geq t_0) + \Pr(t \leq -t_0).$$

But symmetry of the t-distribution around its mean of 0 implies that:

$$\Pr(t \leq -t_0) = \Pr(t \geq t_0).$$

Therefore,

$$\begin{aligned} \text{two-tail p-value of } t_0 &= \Pr(|t| \geq |t_0|) = \Pr(t \geq t_0) + \Pr(t \leq -t_0) \\ &= 2 \cdot \Pr(t \geq t_0). \end{aligned}$$

- To compute the **two-tail p-value** of a calculated t-statistic t_0 that equals 2.0 and has the t-distribution with 60 degrees of freedom, use the **ttail(df, t₀)** function with **df = 60** and **t₀ = 2.0**. Enter the command:

```
display 2*ttail(60, 2.0)
```

Example 1: Two-tail t-tests

Suppose that sample size $N = 62$ and $K = 2$, so that the degrees of freedom for t-tests based on a simple linear regression with two regression coefficients equal $N - K = 62 - 2 = 60$.

- The following are the **two-tail critical values** $t_{\alpha/2}[60]$ of the $t[60]$ distribution, where α is the chosen significance level for the two-tail t-test; they are taken from a published table of percentage points of the t distribution.

$$\begin{aligned} \alpha = 0.01 &\Rightarrow \alpha/2 = 0.005: & t_{\alpha/2}[60] &= t_{0.005}[60] &= 2.660; \\ \alpha = 0.02 &\Rightarrow \alpha/2 = 0.01: & t_{\alpha/2}[60] &= t_{0.01}[60] &= 2.390; \\ \alpha = 0.05 &\Rightarrow \alpha/2 = 0.025: & t_{\alpha/2}[60] &= t_{0.025}[60] &= 2.000; \\ \alpha = 0.10 &\Rightarrow \alpha/2 = 0.05: & t_{\alpha/2}[60] &= t_{0.05}[60] &= 1.671. \end{aligned}$$

- The following commands use the **invttail(df, p)** statistical function to display these **two-tail critical values of the t[60] distribution** at the four chosen significance levels α , namely $\alpha = 0.01, 0.02, 0.05$, and 0.10 :

```
display invttail(60, 0.005)
display invttail(60, 0.01)
display invttail(60, 0.025)
display invttail(60, 0.05)
```

- Now use the **ttail**(*df*, *t*₀) statistical function to display the **two-tail p-values** of the four sample values *t*₀ = 2.660, 2.390, 2.000, and 1.671, which you already know equal the corresponding values of α (0.01, 0.02, 0.05, and 0.10):

```
display 2*ttail(60, 2.660)
display 2*ttail(60, 2.390)
display 2*ttail(60, 2.000)
display 2*ttail(60, 1.671)
```

Note that to compute the **two-tail p-values** of the calculated t-statistics, the values of the **ttail**(*df*, *t*₀) function must be multiplied by 2.

This example demonstrates the relationship between the two statistical functions **ttail**(*df*, *t*₀) and **invttail**(*df*, *p*) for the t-distribution.

*Example 2: Two-tail t-tests when *t*₀ is negative*

- Suppose the sample t-values are *negative*, rather than *positive*. For example, consider the sample t-values *t*₀ = -2.660, and -2.000; you already know that for the t[60] distribution their two-tail p-values are, respectively, 0.01, and 0.05.
- There are (at least) two alternative ways of using the **ttail**(*df*, *t*₀) statistical function to compute the correct **two-tail p-values** for **negative values of *t*₀**. To illustrate, enter the following **display** commands:

```
display 2*(1 - ttail(60, -2.660))
display 2*ttail(60, abs(-2.660))

display 2*(1 - ttail(60, -2.000))
display 2*ttail(60, abs(-2.000))
```

Note the use of the *Stata* absolute value operator **abs()**.

◆ **General recommendation for computing *two-tail p-values* of *t-statistics***

Let t_0 be any calculated sample value of a t-statistic that is distributed under the null hypothesis as a $t[df]$ distribution, where t_0 may be either positive or negative.

The following command will always display the correct *two-tail p-value* of t_0 :

```
display 2*ttail(df, abs(t0))
```

Example 3: Two-tail t-tests

- Use the following **display** commands with the **ttail(df, t₀)** function to obtain the *two-tailed p-values for the t-ratios* of the coefficient estimates from OLS estimation of equation (1). Note that since $N = 74$ and $K = 2$, the appropriate degrees of freedom equal $N - K = 74 - 2 = 72$. Enter the commands:

```
regress price weight  
display 2*ttail(72, abs(5.424))  
display 2*ttail(72, abs(-0.006))  
display 2*ttail(72, 0.006)
```

- *Two-tailed critical values of the t[72] distribution* can be obtained using the following **display** commands with the **invttail(df, p)** statistical function. Note that the value of the argument p equals $\alpha/2$, where α is the significance level chosen for the *two-tail test*. Enter the commands:

```
display invttail(72, 0.005)      (for  $\alpha = 0.01$ ,  $\alpha/2 = 0.005$ )  
display invttail(72, 0.025)   (for  $\alpha = 0.05$ ,  $\alpha/2 = 0.025$ )  
display invttail(72, 0.05)    (for  $\alpha = 0.10$ ,  $\alpha/2 = 0.05$ )
```

Example 4: Two-tail t-tests

Now use the `ttail(df, t0)` function to compute and display the *two-tailed p-values* of the calculated t-statistics `tb0eq0`, `tb1eq0`, `tb1eq1`, and `tb1eq2` for the four two-tail t-tests in the previous section. Recall that the degrees of freedom for these t-statistics equals $N - K = 74 - 2 = 72$, which is the degrees of freedom for the residual sum-of-squares from OLS estimation of equation (1), previously saved as the scalar `dfRSS`. Note that the arguments *df* and *t₀* of the `ttail(df, t0)` function may be either literal numbers (such as 72 and 5.424) or previously saved scalars (such as `dfRSS` and `tb1eq0`).

- **Test 1: a two-tail t-test of $H_0: \beta_1 = 0$ vs. $H_1: \beta_1 \neq 0$**

Enter the commands:

```
scalar list dfRSS tb1eq0
display 2*ttail(72, abs(tb1eq0))
display 2*ttail(dfRSS, abs(tb1eq0))
```

- **Test 2: a two-tail t-test of $H_0: \beta_0 = 0$ vs. $H_1: \beta_0 \neq 0$**

Enter the commands:

```
scalar list dfRSS tb0eq0
display 2*ttail(72, abs(tb0eq0))
display 2*ttail(dfRSS, abs(tb0eq0))
```

- **Test 3: a two-tail t-test of $H_0: \beta_1 = 1$ vs. $H_1: \beta_1 \neq 1$**

Enter the commands:

```
scalar list dfRSS tb1eq1
display 2*ttail(72, abs(tb1eq1))
display 2*ttail(dfRSS, abs(tb1eq1))
```

- **Test 4: a two-tail t-test of $H_0: \beta_1 = 2$ vs. $H_1: \beta_1 \neq 2$**

Enter the commands:

```
scalar list dfRSS tbleq2
display 2*ttail(72, abs(tbleq2))
display 2*ttail(dfRSS, abs(tbleq2))
```

Which of the null hypotheses in Tests 1 to 4 would you reject at the 5 percent significance level? Which would you reject at the 1 percent significance level?

□ Computing a Two-Tail F-test of $H_0: \beta_1 = 0$ vs. $H_1: \beta_1 \neq 0$ -- scalar

This section demonstrates how to use **scalar** commands to compute a *two-tail F-test* of the null hypothesis $H_0: \beta_1 = 0$ against the two-sided alternative hypothesis $H_1: \beta_1 \neq 0$. This is Test 1 from the previous sections.

The **sample value** of the required F-statistic can be computed using either of two equivalent formulas:

$$F_0 = \frac{\hat{\beta}_1^2}{\widehat{\text{Var}}(\hat{\beta}_1)} \sim F[1, N - 2] \text{ under } H_0: \beta_1 = 0; \quad (9.1)$$

$$F_0 = \frac{\text{ESS}/df_{\text{ESS}}}{\text{RSS}/df_{\text{RSS}}} = \frac{\text{ESS}/1}{\text{RSS}/(N - 2)} \sim F[1, N - 2] \text{ under } H_0: \beta_1 = 0. \quad (9.2)$$

- To compute and display the value of F_0 using formula (9.1), enter the commands:

```
scalar Frb1 = b1^2/varb1
scalar list b1 varb1 Frb1
```

- To compute and display the value of F_0 using formula (9.2), enter the commands:

```
scalar F0 = (ESS/dfESS)/(RSS/dfRSS)
scalar list ESS dfESS RSS dfRSS F0
```

- Compare the values of F_0 computed using formulas (9.1) and (9.2) with each other and with the value of the ANOVA F-statistic **Fstat** computed by the **regress** command. Enter the command:

```
scalar list Frb1 F0 Fstat
```

As expected, the sample values of these three F-statistics are equal!

□ Computing a Two-Tail F-test of $H_0: \beta_1 = 0$ vs. $H_1: \beta_1 \neq 0$ – test

An alternative way of computing F-tests of linear coefficient restrictions is to use the *Stata* **test** command. The **test** command is a post-estimation *Stata* command that computes F-tests of linear equality restrictions on regression coefficients. This section demonstrates how to use the **test** command to perform a *two-tail F-test* of Test 1, for which the null and alternative hypotheses are:

$$H_0: \beta_1 = 0 \quad \text{versus} \quad H_1: \beta_1 \neq 0.$$

- To calculate an F-test of this hypothesis, enter the following **test** command:

```
test weight = 0
```

- An equivalent **test** command omits the "= 0" part of the command. Enter the command:

```
test weight
```

- The **test** command temporarily stores results in **r()** functions; the calculated sample value of the F-statistic is stored in **r(F)**. To list all temporarily saved results from a **test** command, enter the command:

```
return list
```

- To name and save the F-statistic computed by the preceding **test** command, enter the **scalar** command:

```
scalar Ftest = r(F)
```

- Compare the sample value of the F-statistic calculated by the **test** command with the previously computed values of the same F-statistic. Enter the command:

```
scalar list Frb1 F0 Fstat Ftest
```

- The *sample value of the F-statistic* and the *sample value of the t-statistic* for testing $H_0: \beta_1 = 0$ against $H_1: \beta_1 \neq 0$ are related as follows:

$$F_0 = \frac{\hat{\beta}_1^2}{\text{Var}(\hat{\beta}_1)} = \left(\frac{\hat{\beta}_1}{\sqrt{\text{Var}(\hat{\beta}_1)}} \right)^2 = \left(\frac{\hat{\beta}_1}{\text{s}\hat{\text{e}}(\hat{\beta}_1)} \right)^2 = (t_0)^2. \quad (10)$$

As equation (10) shows, the sample value F_0 of the F-statistic for testing $H_0: \beta_1 = 0$ against $H_1: \beta_1 \neq 0$ equals the *square* of the sample value t_0 of the corresponding t-statistic. To verify that this is so, enter the following commands:

```
scalar tbleq0sq = tbleq0^2
scalar list tbleq0sq Frb1 F0 Fstat Ftest
```

□ Computing critical values of F-distributions and p-values for F-statistics

Basic Syntax: The *Stata 10* statistical functions for the F-distribution are **Ftail**(df_1 , df_2 , F_0) and **invFtail**(df_1 , df_2 , p).

- **Ftail**(df_1 , df_2 , F_0) computes the *right-tail (upper-tail) p-value of an F-statistic* that has df_1 *numerator* degrees of freedom, df_2 *denominator* degrees of freedom, and calculated sample value F_0 . It returns the probability that $F > F_0$, i.e., the value of $\Pr(F > F_0)$.
- **invFtail**(df_1 , df_2 , p) computes the *right-tail critical value of an F-distribution* with df_1 *numerator* degrees of freedom, df_2 *denominator* degrees of freedom, and *probability level* p . If α denotes the chosen significance level for the F-test, then set $p = \alpha$.
- If **Ftail**(df_1 , df_2 , F_0) = p , then **invFtail**(df_1 , df_2 , p) = F_0 .

Usage: The statistical functions **Ftail**(df_1 , df_2 , F_0) and **invFtail**(df_1 , df_2 , p) must be used with *Stata 10* commands such as **display**, **generate**, **replace**, or **scalar**; they cannot be used by themselves. For example, simply typing **Ftail(2, 60, 4.69)** will produce an error message. Instead, to obtain the p-value for a calculated F-statistic that equals 4.69 and has the F-distribution with 2 numerator degrees of freedom and 60 denominator degrees of freedom, enter the **display** command:

```
display Ftail(2, 60, 4.69)
```

Similarly, to obtain the 5 percent critical value of the F[2, 60] distribution, type

```
display invFtail(2, 60, 0.05)
```

Example 1

- The following are the **critical values** $F_{\alpha}[4, 60]$ of the F[4, 60] distribution, where α is the chosen significance level for the F-test; they are taken from a published table of upper percentage points of the F-distribution.

$$\alpha = 0.01: \quad F_{\alpha}[4, 60] = F_{0.01}[4, 60] = 3.649;$$

$$\alpha = 0.05: \quad F_{\alpha}[4, 60] = F_{0.05}[4, 60] = 2.525.$$

- The following commands use the **invFtail**(df_1 , df_2 , p) statistical function to display these **critical values of the F[4, 60] distribution** at the two chosen significance levels α , namely $\alpha = 0.01$ and 0.05 :

```
display invFtail(4, 60, 0.01)
```

```
display invFtail(4, 60, 0.05)
```

- From tables of critical values of the F-distribution in a textbook, you can obtain the following critical values of the F[2, 60] distribution for the four significance levels $\alpha = 0.01$, 0.05 , 0.10 , and 0.25 :

$$\text{for } \alpha = 0.01, \quad F_{0.01}[2, 60] = 4.98;$$

$$\text{for } \alpha = 0.05, \quad F_{0.05}[2, 60] = 3.15;$$

$$\text{for } \alpha = 0.10, \quad F_{0.10}[2, 60] = 2.39;$$

$$\text{for } \alpha = 0.25, \quad F_{0.25}[2, 60] = 1.42.$$

The following commands use the **invFtail**(df_1 , df_2 , α) statistical function to display these *critical values* of the F[2, 60] distribution:

```
display invFtail(2,60, 0.01)
display invFtail(2,60, 0.05)
display invFtail(2,60, 0.10)
display invFtail(2,60, 0.25)
```

- Now use the **Ftail**(df_1 , df_2 , F_0) statistical function to display the *p-values* of the four critical values 4.98, 3.15, 2.39, and 1.42 of the F[2,60] distribution, which you already know equal 0.01, 0.05, 0.10, and 0.25, respectively:

```
display Ftail(2,60, 4.98)
display Ftail(2,60, 3.15)
display Ftail(2,60, 2.39)
display Ftail(2,60, 1.42)
```

Example 2

- Use the following **display** commands with the **Ftail**(df_1 , df_2 , F_0) function to compute the **p-values** of the previously calculated F-statistics for testing $H_0: \beta_1 = 0$ against $H_1: \beta_1 \neq 0$. Enter the commands:

```
display Ftail(1,72,Frb1)
display Ftail(dfESS, dfRSS, F0)
display Ftail(1, 72, Fstat)
display Ftail(dfESS, dfRSS, Ftest)
```

Note the alternative ways of specifying the numerator and denominator degrees of freedom in the **Ftail**(df_1 , df_2 , F_0) function; they can be specified either as literal numbers (positive integers such as 1 and 72) or as pre-defined scalars (such as **dfESS** and **dfRSS**).

- Compare the *p-value of the calculated F-statistic* with the *two-tailed p-value of the calculated t-statistic* for testing $H_0: \beta_1 = 0$ against $H_1: \beta_1 \neq 0$. Enter the commands:

```
display 2*ttail(72, abs(tbleq0))
display 2*ttail(dfRSS, abs(tbleq0))
```

- Finally, calculate and compare the **two-tailed 1% critical value** of the **t[72]**-distribution and the **1% critical value** of the **F[1, 72]**-distribution. Enter the **display** commands:

```
display invttail(dfrSS, 0.005)
display invFtail(dfESS, dfrSS, 0.01)
```

The results of these two commands can be used to verify that the **α -level critical value of the F[1, 72]-distribution equals the square of the $\alpha/2$ critical value of the t[72] distribution**, i.e., that

$$F_{0.01}[1, 72] = (t_{0.005}[72])^2$$

or in general that

$$F_{\alpha}[1, 72] = (t_{\alpha/2}[72])^2 \text{ for significance level } \alpha.$$

□ **Computing One-Tail t-tests and p-values -- invttail(,), ttail(,)**

Students of introductory econometrics often find **one-tail hypothesis tests** particularly difficult. But such tests are important in applied econometric research because the predictions of economic theory frequently take the form of predictions about the **direction of relationships** between economic variables.

We begin with some simple examples of how to use the *Stata* statistical functions **ttail(df, t₀)** and **invttail(df, p)** in performing one-tail t-tests. Recall that there are two categories of one-tail t-tests: **right-tail t-tests**; and **left-tail t-tests**. For one-tail t-tests, the parameter **p** of the **invttail(df, p)** statistical function is set equal to the chosen significance level **α** : i.e., for one-tail t-tests, set **p = α** .

Example 1: One-tail t-tests -- right-tail t-tests

- The following are the **upper one-tail (right-tail) critical values $t_{\alpha}[72]$** of the $t[72]$ distribution, where α is the chosen significance level for the one-tail t-test; they are computed from a published table of percentage points of the t-distribution.

$$\alpha = 0.01: \quad t_{\alpha}[72] = t_{0.01}[72] = 2.379;$$

$$\alpha = 0.05: \quad t_{\alpha}[72] = t_{0.05}[72] = 1.666;$$

$$\alpha = 0.10: \quad t_{\alpha}[72] = t_{0.10}[72] = 1.293.$$

- The following commands use the **invttail(df, p)** statistical function with $p = \alpha$ to display these **upper one-tail (right-tail) critical values of the $t[72]$ distribution** at the 1%, 5% and 10% significance levels, i.e., for $\alpha = 0.01, 0.05$ and 0.10:

```
display invttail(72, 0.01)
```

```
display invttail(72, 0.05)
```

```
display invttail(72, 0.10)
```

- Now use the **ttail(df, t_0)** statistical function to display the **right-tail p-values** of the sample t-values $t_0 = 2.379, 1.666,$ and $1.293,$ which you already know equal the corresponding values of $\alpha,$ namely 0.01, 0.05, and 0.10, respectively:

```
display ttail(72, 2.379)
```

```
display ttail(72, 1.666)
```

```
display ttail(72, 1.293)
```

Example 2: One-tail t-tests -- left-tail t-tests

- The following are the **lower one-tail (left-tail) critical values $-t_{\alpha}[72]$** of the $t[72]$ distribution, where α is the chosen significance level for the left-tail t-test; they are simply equal to the **negative** of the corresponding **upper one-tail (right-tail) critical values $t_{\alpha}[72]$** taken from a published table of percentage points of the t distribution.

$$\alpha = 0.01: \quad -t_{\alpha}[72] = -t_{0.01}[72] = -2.379;$$

$$\alpha = 0.05: \quad -t_{\alpha}[72] = -t_{0.05}[72] = -1.666;$$

$$\alpha = 0.10: \quad -t_{\alpha}[72] = -t_{0.10}[72] = -1.293.$$

- The following commands use the **invttail(df, p)** statistical function to display the **lower one-tail (left-tail) critical values of the t[72] distribution** at the 1%, 5% and 10% significance levels, i.e., for $\alpha = 0.01, 0.05$ and 0.10 :

```
display -1*invttail(72, 0.01)
display -1*invttail(72, 0.05)
display -1*invttail(72, 0.10)
```

- Now use the **ttail(df, t₀)** statistical function to display the **left-tail p-values** of the sample t-values $t_0 = -2.390, -1.671,$ and $-1.296,$ which you already know equal the corresponding values of $\alpha,$ namely $0.01, 0.05,$ and $0.10,$ respectively:

```
display 1 - ttail(72, -2.379)
display 1 - ttail(72, -1.666)
display 1 - ttail(72, -1.293)
```

□ Computing One-Tail t-tests and p-values -- scalar, invttail(,), ttail(,)

This section gives you some practice in computing **one-tail t-tests** and the **p-values** of the calculated t-statistics. It uses **scalar** commands to compute the sample values of the t-statistic. Recall that the **feasible t-statistic** for $\hat{\beta}_1$ takes the general form:

$$t(\hat{\beta}_1) = \frac{\hat{\beta}_1 - \beta_1}{\widehat{se}(\hat{\beta}_1)} \sim t[N - 2]. \quad (2)$$

Test 5: $H_0: \beta_1 = 0$ or $\beta_1 \leq 0$
 $H_1: \beta_1 > 0$ a **one-sided** alternative \Rightarrow a **one-tail right tail test**

- Calculate the **sample value of the t-statistic** (2) under the null hypothesis H_0 . In performing Test 1 above, you previously computed the value of t-statistic (2) under the equality form of the null hypothesis -- i.e., for $\beta_1 = 0$. To redisplay these results, enter the command:

```
scalar list b1 seb1 tbleq0
```

These commands calculate and display the following sample value t_0 of the t-statistic for $\hat{\beta}_1$:

$$t_0(\hat{\beta}_1) = t_0 = \frac{\hat{\beta}_1 - \beta_1}{\widehat{\text{se}}(\hat{\beta}_1)} = \frac{2.0441 - 0}{0.37683} = \frac{2.0441}{0.37683} = \mathbf{5.424}.$$

- Calculate $t_\alpha[N-2]$, the **upper α -level critical value** of the $t[N-2] = t[72]$ distribution, for $\alpha = 0.05$ and $\alpha = 0.01$. Use the **invttail(df, p)** function to do this. For each value of α , calculate the appropriate value of **p** in the **invttail (df, p)** function as follows:

$$p = \alpha$$

For $\alpha = 0.05$: $p = \alpha = 0.05$.

For $\alpha = 0.01$: $p = \alpha = 0.01$.

- Enter the following **display** commands:

```
display invttail(72, 0.05)
display invttail(72, 0.01)
```

The first **display** command returns the value of the **upper (right-tail) 5 percent critical value** = $t_\alpha[72] = t_{0.05}[72] = 1.666$.

The second **display** command returns the value of the **upper (right-tail) 1 percent critical value** = $t_\alpha[72] = t_{0.01}[72] = 2.379$.

- **Decision Rule -- Formulation 1**: At significance level α ,

- ◆ **reject H_0** if $t_0(\hat{\beta}_1) > t_\alpha[72]$;

- ◆ **retain H_0** if $t_0(\hat{\beta}_1) \leq t_\alpha[72]$.

The calculated **sample value of the t-statistic** is $t_0(\hat{\beta}_1) = t_0 = \mathbf{5.424}$.

At the **5% significance level**, $t_\alpha[72] = t_{0.05}[72] = 1.666$.

$$t_0 = 5.424 > 1.666 = t_{0.05}[72] \Rightarrow \mathbf{reject} H_0: \beta_1 = 0 \text{ vs } H_1: \beta_1 > 0 \text{ at the } \mathbf{5\%} \text{ significance level.}$$

At the **1% significance level**, $t_{\alpha}[72] = t_{0.01}[72] = 2.379$.

$$t_0 = 5.424 > 2.379 = t_{0.01}[72] \Rightarrow \text{reject } H_0: \beta_1 = 0 \text{ vs } H_1: \beta_1 > 0 \\ \text{at the 1\% significance level.}$$

Inference: *Reject* the null hypothesis $H_0: \beta_1 = 0$ against the alternative hypothesis $H_1: \beta_1 > 0$ at *both* the **5%** and **1%** significance levels.

➤ **Decision Rule -- Formulation 2:** At significance level α ,

- ◆ *reject* H_0 if *right-tail p-value* for $t_0 < \alpha$;
- ◆ *retain* H_0 if *right-tail p-value* for $t_0 \geq \alpha$.

Note: Because this is a *right-tail test*, we want to compute the *right-tail p-value* for t_0 where

$$\text{right-tail p-value for } t_0 = \Pr(t > t_0).$$

Use the `ttail(df, t0)` function to compute the *right-tail p-value* for t_0 . Since $t_0 = 5.424 > 0$,

$$\text{ttail}(df, 5.424) = \Pr(t > 5.424).$$

- Thus, the desired *right-tail p-value* for t_0 can be obtained by entering the command:

```
display ttail(72, 5.424)
```

This command returns the value $3.713e-07 = 0.0000003713 = 0.0000$.

At **5% significance level**, $\alpha = 0.05$:

$$\text{right-tail p-value for } t_0 = 0.0000 < 0.05 \Rightarrow \text{reject } H_0: \beta_1 = 0 \text{ vs. } H_1: \beta_1 > 0 \\ \text{at the 5\% significance level.}$$

At **1% significance level**, $\alpha = 0.01$:

right-tail p-value for $t_0 = 0.0000 < 0.01 \Rightarrow$ **reject** $H_0: \beta_1 = 0$ vs. $H_1: \beta_1 > 0$
at the **1%** significance level.

Inference: **Reject** the null hypothesis $H_0: \beta_1 = 0$ against the alternative hypothesis $H_1: \beta_1 > 0$ at **both** the **5%** and **1%** significance levels.

Test 6: $H_0: \beta_1 = 0$ or $\beta_1 \geq 0$
 $H_1: \beta_1 < 0$ a **one-sided** alternative \Rightarrow a **one-tail left tail test**

- You have previously calculated the **sample value of the t-statistic** (2) under the equality form of H_0 , that is at $\beta_1 = 0$. To redisplay the results, enter the command:

```
scalar list b1 seb1 tbleq0
```

The sample value t_0 of the t-statistic for $\hat{\beta}_1$ under H_0 is calculated as:

$$t_0(\hat{\beta}_1) = t_0 = \frac{\hat{\beta}_1 - \beta_1}{s\hat{e}(\hat{\beta}_1)} = \frac{2.0441 - 0}{0.37683} = \frac{2.0441}{0.37683} = \mathbf{5.424}.$$

- Calculate $-t_\alpha[N-2]$, the **lower α -level critical value** of the $t[N-2] = t[72]$ distribution, for $\alpha = 0.05$ and $\alpha = 0.01$.

Enter the following **display** commands:

```
display -invttail(72, 0.05)
display -invttail(72, 0.01)
```

The first **display** command returns the value of the **lower (left-tail) 5 percent critical value** $= -t_\alpha[72] = -t_{0.05}[72] = -1.666$.

The second **display** command returns the value of the **lower (left-tail) 1 percent critical value** $= -t_\alpha[72] = -t_{0.01}[72] = -2.379$.

➤ **Decision Rule -- Formulation 1:** At significance level α ,

- ◆ *reject* H_0 if $t_0(\hat{\beta}_1) < -t_\alpha[72]$;
- ◆ *retain* H_0 if $t_0(\hat{\beta}_1) \geq -t_\alpha[72]$.

The calculated **sample value of the t-statistic** is $t_0(\hat{\beta}_1) = t_0 = 5.424$.

At the **5% significance level**, $-t_\alpha[72] = -t_{0.05}[72] = -1.666$.

$$t_0 = 5.424 > -1.666 = -t_{0.05}[72] \Rightarrow \textit{retain } H_0: \beta_1 = 0 \text{ vs } H_1: \beta_1 < 0 \\ \text{at the } \mathbf{5\%} \text{ significance level.}$$

At the **1% significance level**, $-t_\alpha[72] = -t_{0.01}[72] = -2.379$.

$$t_0 = 5.424 > -2.379 = -t_{0.01}[72] \Rightarrow \textit{retain } H_0: \beta_1 = 0 \text{ vs } H_1: \beta_1 < 0 \\ \text{at the } \mathbf{1\%} \text{ significance level.}$$

Inference: *Retain* the null hypothesis $H_0: \beta_1 = 0$ against the alternative hypothesis $H_1: \beta_1 < 0$ at *both* the **5%** and **1%** significance levels.

➤ **Decision Rule -- Formulation 2:** At significance level α ,

- ◆ *reject* H_0 if the *left-tail p-value* for $t_0 < \alpha$;
- ◆ *retain* H_0 if the *left-tail p-value* for $t_0 \geq \alpha$.

Note: Because this is a *left-tail test*, we want to compute the *left-tail p-value* for t_0 where

$$\textit{left-tail p-value for } t_0 = \Pr(t < t_0).$$

Use the **ttail(df, t₀)** function to compute the *left-tail p-value* for t_0 . Recall that **ttail(df, t₀)** returns the *right-tail p-value* for t_0 . Since $t_0 = 5.424 > 0$,

$$\text{ttail}(df, 5.424) = \Pr(t > 5.424).$$

- But we want the $\Pr(t < t_0) = \Pr(t < 5.424) = 1 - \Pr(t > 5.424)$. Thus, the desired **left-tail p-value for t_0** can be obtained by entering the command:

```
display 1 - ttail(72, 5.424)
```

This command returns the value $0.99999963 = 0.9999$.

At the **5% significance level**, $\alpha = 0.05$:

left-tail p-value for $t_0 = 0.9999 > 0.05 \Rightarrow$ **retain** $H_0: \beta_1 = 0$ vs $H_1: \beta_1 < 0$
at the **5%** significance level.

At the **1% significance level**, $\alpha = 0.01$:

left-tail p-value for $t_0 = 0.9999 > 0.01 \Rightarrow$ **retain** $H_0: \beta_1 = 0$ vs $H_1: \beta_1 < 0$
at the **1%** significance level.

Inference: **Retain** the null hypothesis $H_0: \beta_1 = 0$ against the alternative hypothesis $H_1: \beta_1 < 0$ at **both** the **5%** and **1%** significance levels.

Test 7: $H_0: \beta_1 = 1$ or $\beta_1 \leq 1$
 $H_1: \beta_1 > 1$ a **one-sided** alternative \Rightarrow a **one-tail right tail test**

- Calculate the **sample value of the t-statistic** (2) under the null hypothesis H_0 . Use the equality form of H_0 ; that is, set $\beta_1 = 1$ in (2). This yields the following sample value t_0 of the t-statistic for $\hat{\beta}_1$ under $H_0: \beta_1 = 1$:

$$t_0(\hat{\beta}_2) = t_0 = \frac{\hat{\beta}_2 - \beta_2}{\text{s}\hat{\text{e}}(\hat{\beta}_2)} = \frac{2.0441 - 1}{0.37683} = \frac{1.0441}{0.37683} = \mathbf{2.771}.$$

In performing Test 3 above, you previously calculated the above *sample value* of t-statistic (2) at $\beta_1 = 1$. To redisplay the results of these previous calculations, enter the commands:

```
display b1 - 1
scalar list b1 seb1 tbleq1
```

- Calculate $t_{\alpha}[N-2]$, the **upper α -level critical value** of the $t[N-2] = t[72]$ distribution, for $\alpha = 0.05$ and $\alpha = 0.01$.

Re-enter the following **display** commands:

```
display invttail(72, 0.05)
display invttail(72, 0.01)
```

➤ **Decision Rule -- Formulation 1:** At significance level α ,

- ◆ **reject H_0** if $t_0(\hat{\beta}_1) > t_{\alpha}[72]$;
- ◆ **retain H_0** if $t_0(\hat{\beta}_1) \leq t_{\alpha}[72]$.

The calculated **sample value of the t-statistic** is $t_0(\hat{\beta}_1) = t_0 = 2.771$.

At the **5% significance level**, $t_{\alpha}[72] = t_{0.05}[72] = 1.666$.

$$t_0 = 2.771 > 1.666 = t_{0.05}[72] \Rightarrow \text{reject } H_0: \beta_1 = 1 \text{ vs. } H_1: \beta_1 > 1 \\ \text{at the 5\% significance level.}$$

At the **1% significance level**, $t_{\alpha}[72] = t_{0.01}[72] = 2.379$.

$$t_0 = 2.771 > 2.379 = t_{0.01}[72] \Rightarrow \text{reject } H_0: \beta_1 = 1 \text{ vs. } H_1: \beta_1 > 1 \\ \text{at the 1\% significance level.}$$

Inference: *Reject* the null hypothesis $H_0: \beta_1 = 1$ against the alternative hypothesis $H_1: \beta_1 > 1$ at **both** the **5%** and **1%** significance levels.

➤ **Decision Rule -- Formulation 2:** At significance level α ,

- ◆ **reject H_0** if **right-tail p-value** for $t_0 < \alpha$;
- ◆ **retain H_0** if **right-tail p-value** for $t_0 \geq \alpha$.

Note: Because this is a *right-tail test*, we want to compute the *right-tail p-value* for t_0 where

$$\textit{right-tail p-value for } t_0 = \Pr(t > t_0).$$

Since the `ttail(df, t0)` function computes the *right-tail p-value* for t_0 ,

$$\text{ttail}(df, 2.771) = \Pr(t > 2.771).$$

- Thus, the desired *right-tail p-value* for $t_0 = 2.771$ can be obtained by entering the command:

```
display ttail(72, 2.771)
```

This command returns the value $0.00355314 = 0.003553$.

At the **5% significance level**, $\alpha = 0.05$:

right-tail p-value for $t_0 = 0.003553 < 0.05 \Rightarrow$ *reject* $H_0: \beta_1 = 1$ vs. $H_1: \beta_1 > 1$
at the **5%** significance level.

At the **1% significance level**, $\alpha = 0.01$:

right-tail p-value for $t_0 = 0.003553 < 0.01 \Rightarrow$ *reject* $H_0: \beta_1 = 1$ vs. $H_1: \beta_1 > 1$
at the **1%** significance level.

Inference: *Reject* the null hypothesis $H_0: \beta_1 = 1$ against the alternative hypothesis $H_1: \beta_1 > 1$ at *both* the **5%** and **1%** significance levels.

Test 8: $H_0: \beta_1 = 3$ or $\beta_1 \geq 3$
 $H_1: \beta_1 < 3$ a **one-sided** alternative \Rightarrow a **one-tail left tail test**

- Calculate the *sample value of the t-statistic* (2) under the null hypothesis H_0 . Use the equality form of H_0 ; that is, set $\beta_1 = 3$ in (2). Enter the commands:

```
display b1 - 3
scalar tbleq3 = (b1 - 3)/seb1
scalar list b1 seb1 tbleq3
```

These commands calculate and display the following sample value t_0 of the t-statistic for $\hat{\beta}_1$:

$$t_0(\hat{\beta}_1) = t_0 = \frac{\hat{\beta}_1 - \beta_1}{\widehat{se}(\hat{\beta}_1)} = \frac{2.0441 - 3}{0.37683} = \frac{-0.9559}{0.37683} = -2.537.$$

- Calculate $-t_{\alpha}[N-2]$, the **lower α -level critical value** of the $t[N-2] = t[72]$ distribution, for $\alpha = 0.05$ and $\alpha = 0.01$.

Re-enter the following **display** commands:

```
display -invttail(72, 0.05)
display -invttail(72, 0.01)
```

➤ **Decision Rule -- Formulation 1:** At significance level α ,

- ◆ **reject H_0** if $t_0(\hat{\beta}_1) < -t_{\alpha}[72]$;
- ◆ **retain H_0** if $t_0(\hat{\beta}_1) \geq -t_{\alpha}[72]$.

The calculated *sample value of the t-statistic* is $t_0(\hat{\beta}_1) = t_0 = -2.537$.

At the **5% significance level**, $-t_{\alpha}[72] = -t_{0.05}[72] = -1.666$.

$$t_0 = -2.537 < -1.666 = -t_{0.05}[72] \Rightarrow \text{reject } H_0: \beta_1 = 3 \text{ vs } H_1: \beta_1 < 3$$

at the **5%** significance level.

At the **1% significance level**, $-t_{\alpha}[72] = -t_{0.01}[72] = -2.379$.

$$t_0 = -2.537 < -2.379 = -t_{0.01}[72] \Rightarrow \text{reject } H_0: \beta_1 = 3 \text{ vs } H_1: \beta_1 < 3 \\ \text{at the } \mathbf{1\%} \text{ significance level.}$$

Inference: *Reject* the null hypothesis $H_0: \beta_1 = 3$ against the alternative hypothesis $H_1: \beta_1 < 3$ at *both* the **5%** and **1%** significance levels.

➤ **Decision Rule -- Formulation 2:** At significance level α ,

- ◆ *reject* H_0 if the *left-tail p-value* for $t_0 < \alpha$;
- ◆ *retain* H_0 if the *left-tail p-value* for $t_0 \geq \alpha$.

Note: Because this is a *left-tail test*, we want to compute the *left-tail p-value* for t_0 where

$$\textit{left-tail p-value for } t_0 = \Pr(t < t_0).$$

Use the `ttail(df, t0)` function to compute the *left-tail p-value* for t_0 . Recall that the *Stata* statistical function `ttail(df, t0)` returns the *right-tail p-value* for t_0 , so that

$$\text{ttail}(df, -2.537) = \Pr(t > -2.537).$$

The *left-tail p-value* for t_0 can therefore be obtained by subtracting the above p-value from 1:

$$1 - \text{ttail}(df, -2.537) = 1 - \Pr(t > -2.537) = \Pr(t < -2.537).$$

- Thus, the desired *left-tail p-value* for t_0 can be obtained by entering the command:

```
display 1 - ttail(72, -2.537)
```

This command returns the value $0.00667398 = 0.006674$.

At the **5% significance level**, $\alpha = 0.05$:

left-tail p-value for $t_0 = 0.006674 < 0.05 \Rightarrow$ **reject** $H_0: \beta_1 = 3$ vs $H_1: \beta_1 < 3$
at the **5%** significance level.

At the **1% significance level**, $\alpha = 0.01$:

left-tail p-value for $t_0 = 0.006674 < 0.01 \Rightarrow$ **reject** $H_0: \beta_1 = 3$ vs $H_1: \beta_1 < 3$
at the **1%** significance level.

Inference: **Reject** the null hypothesis $H_0: \beta_1 = 3$ against the alternative hypothesis $H_1: \beta_1 < 3$ at **both** the **5%** and **1%** significance levels.

□ Preparing to End Your *Stata* Session

Before you end your *Stata* session, you should do two things.

- First, you may want to **save the current dataset** (although you will not need it for future tutorials). Enter the following **save** command to save the current dataset as *Stata*-format dataset **auto5.dta**:

```
save auto5
```

- Second, **close the .log file** you have been recording. Enter the command:

```
log close
```

□ End Your *Stata* Session – exit

- **To end your *Stata* session**, use the **exit** command. Enter the command:

```
exit      or      exit, clear
```

□ Cleaning Up and Clearing Out

After returning to Windows, you should copy all the files you have used and created during your *Stata* session to your own diskette. These files will be found in the *Stata working directory*, which is usually **C:\data** on the computers in Dunning 350, and **D:\courses** on the computers in MC B111. There is one file you will want to be sure you have: the *Stata* log file **351tutorial5.log**. If you saved the *Stata*-format data set **auto5.dta**, you will probably want to take it with you as well. Use the Windows **copy** command to copy any files you want to keep to your own portable electronic storage device (e.g., flash memory stick) in the E:-drive (or to a diskette in the A:-drive).

Finally, **as a courtesy to other users** of the computing classroom, please delete all the files you have used or created from the *Stata* working directory.